

Регулярные типы (массивы)

Что такое массив?

Дана таблица, которая содержит среднемесячные значения температуры в городе в 2010 году:

Месяц	1	2	3	4	5	6	7	8	9	10	11	12
Температура	-19	-18	-8	6	14	18	23	24	16	4	-7	-18

Такую таблицу называют линейной. Она представляет собой последовательность чисел.

Для обозначения этих чисел используют индексированные имена.

T	Месяц	1	2	3	4	5	6	7	8	9	10	11	12
	Температура	-19	-18	-8	6	14	18	23	24	16	4	-7	-18

Например, через **T[1]** обозначается температура в январе (первом месяце года), **T[5]** – температура в мае и т. д.

В программировании линейная таблица называется **одномерным массивом**.

В нашем примере T – это имя массива. Элементы массива пронумерованы. Порядковый номер элемента называется его индексом. Каждый элемент массива обозначается индексированным именем.

Массивом называется упорядоченная последовательность величин одного типа, обозначаемая одним именем.

Примечание

Упорядоченность заключается в том, что элементы массива располагаются в последовательных ячейках памяти.

Особенности:

- весь массив имеет одно имя
- все элементы имеют один тип
- все элементы расположены в памяти рядом

Чтобы получить **доступ** к нужной ячейке, нужно указать имя массива и его индекс (порядковый номер элемента в массиве), который записывается в квадратных скобках после имени массива.

Количество элементов в массиве называется его **размерностью**.

Каждый элемент массива имеет:

1. имя
2. индекс
3. значение.

Описание (объявление) массива

Array [диапазон_индексов] **of** тип_компонентов

Имя переменной:

Задание типа:

Значение:

MAS		
Array[1..3] of Real		
3.23	-2.5	12.74

Элементы массива

1. определить имя массива
2. определить число элементов
3. определить тип массива
4. выделить место в памяти

Имя переменной:

Индекс:

Значение:

Тип элементов:

MAS		
[1]	[2]	[3]
3.23	-2.5	12.74
Real	Real	Real

Элементы массива

Способы описания массива

1 способ	2 способ
Var X:Array[1..3] of Real;	Type Field=Array[1..3] of Real; Var X : Field;

Примеры использования в качестве индексов различных типов**Type**

```

MonthType    = (Januare, Februaire, March, April, May);
ComplectType = Array [ MonthType ] of Word;
SpringType   = Array [March..May ] of Word;

```

Var

```

Complect:ComplectType;           {пять элементов типа Word}
Spring  :SpringType;             {три элемента типа Word}
Alpha   :Array['A'.. 'z'] of Char;
Switch  :Array[Boolean] of Byte; {два элемента типа Byte}
Array_1 :Array[9..99] of Char;    {массив из 91 элемента}
Array_2 :Array[-10..10] of LongInt; {массив из 21 элемента}

```

Последовательность действий при обработке массивов:

1. объявить массив,
2. заполнить массив данными,
3. вывести на экран исходный массив,
4. обработать массив, указанным в задании образом,
5. вывести на экран результаты работы.

Типичные операции над массивами

1. Определение размерности массива Mas[N]

```

Const    N=10;
// при изменении N программа не должна меняться!
Type    T_Mas=Array[1..N] Of Integer;
Var      Mas  :T_Mas;
           Kol  :Byte;
           I    :Integer;
           IOR  :Word;

Begin
Write('Введите размерность массива: ');
Repeat
    {$I-} ReadLn(Kol); {$I+}
    IOR:=IORResult;
    If (IOR<>0) or (Kol>N) Then
        WriteLn('Ошибка. Повторите ввод. ');
Until (Kol<=N) and (IOR=0);
. . .
End.

```

2. Ввод элементов массива

```

For I:=1 To Kol Do
    ReadLn(Mas[I]);

```

3. Заполнение элементов массива случайными числами

```

Randomize; {Заполнение массива данными}
For I:=1 To Kol Do
    Begin
        Mas[I]:=Random(N);
        WriteLn('MAS[' , I:2, ']=' , Mas[I]:4)
    End;

```

4. Поиск

```

Const    N=10;
Type    T_Mas=Array[1..N] Of Integer;
Var     Mas      :T_Mas;
          Kol,NN   :Integer;
          I        :Integer;
          Max, Max1, Max2 :Integer;
          Flag     : Boolean;

Begin
. . .
Write('Введите размерность массива: ');
ReadLn(Kol);

```

4.1. Поиск значения максимального элемента

для элементов массива принадлежащих заданному диапазону (например, [-10, 10])

```

Max:= -11;
For I:=2 To KOL Do
    If Max<Mas[I] Then Max:=Mas[I];

```

для элементов, значения которых неопределено

```

Max:=Mas[1];
For I:=2 To KOL Do
    If Max<Mas[I] Then Max:=Mas[I];

```

4.2. Поиск номера первого максимального элемента массива

```

Max:=Mas[1];N_Max:=1;
For I:=2 To KOL Do
    If Max<Mas[I] Then
        Begin
            Max:=Mas[I];
            N_Max:=I
        End;

```

4.3. Поиск номера последнего максимального элемента массива

```
Max:=Mas[1];N_Max:=1;
For I:=2 To KOL Do
    If Max<= Mas[I] Then
        Begin
            Max:=Mas[I];
            N_Max:=I
        End;
```

4.4. Поиск номера последнего отрицательного элемента

```
K:=0;
For I:=1 To KOL Do
    If Mas[I]<0 Then K:=I;
If K=0 Then WriteLn('Отр.элементов нет')
    Else WriteLn('Номер последнего отр. ',K);
```

А если отрицательный элемент стоит ближе к концу массива, то мы будем делать лишние операции.

Более рациональное решение задачи:

```
I:=KOL;
Repeat
    Dec(I);
Until (I<1) or (Mas[I]<0);
If I<1 Then WriteLn ('Отрицательных элементов нет')
    Else WriteLn ('Номер последнего отр. ',I);
```

4.5. Поиск количества элементов, равных максимальному значению

```

Max:=Mas [1];N_Max:=1;
For I:=2 To KOL Do
  If Max<Mas [I] Then
    Begin Max:=Mas [I]; N_Max:=1 End
  Else
    If Max=Mas [I] Then N_Max:=N_Max+1;

```

4.6. Поиск значения второго после максимального элемента массива

```

If Mas [1]>Mas [2]
  Then begin Max1:=Mas [1]; Max2:=Mas [2] end
  Else begin Max1:=Mas [2]; Max2:=Mas [1] end;
For I:=3 To KOL Do
  If Max1< Mas [I]
    Then
      Begin Max2:=Max1 Max1:=Mas [I] End
    Else
      If Max2< Mas [I] Then Max2:=Mas [I];

```

Правильно ли будет работать этот алгоритм?

А если первые элементы массива равны и принимают максимальное значение?

Более правильное решение задачи:

```

Max1:=Mas [1]; K:=1;
Flag:=True;
While (K<KOL) and Flag do
  begin
    K:=K+1;
    If Max1< Mas [K]
      Then
        Begin Max2:=Max1; Max1:=Mas [K]; Flag:=False End
      Else
        If Max1<> Mas [K] Then
          Begin Max2:=Mas [K]; Flag:=False End

```

```

End;
If not Flag
  Then
    begin
      For I:=K+1 To KOL Do
        begin
          If Max1< Mas[I]
            Then
              Begin Max2:=Max1; Max1:=Mas[I] End
            Else
              If (Max2< Mas[I]) and (Max1<> Mas[I])
                Then Max2:=Mas[I]
              end;
            WriteLn(Max1:5,Max2:5)
          end
        Else WriteLn('Все элементы равны ');

```

4.7. Поиск элемента, равного максимальному отрицательному значению

```

K:=1;
While MAS[K]>=0 do Inc(K);
Max:=Mas[K];
For I:=K+1 to KOL do
  If Mas[I]<0 Then
    If Mas[I]>Max Then Max:=Mas[I];

```

Правильно ли будет работать этот алгоритм?

А если в массиве нет отрицательных элементов?

Более правильное решение задачи:

```
K:=1;
While (K<KOL) and (Mas[K]>=0) do Inc(K);
If Mas[K] < 0 Then
  begin
    Max:=Mas[K];
    For I:=K+1 to KOL do
      If Mas[I]<0 then
        If Mas[I]>Max then Max:=Mas[I];
    Write(Max);
  end
  else write('таких нет');
```

Или более элегантное решение:

```
K:=0;
  For I:=1 to KOL do
    If Mas[I]<0 then
      If (K=0) or (Mas[I]>Mas[K]) Then K:=I;
If K>0 Then Writeln(Mas[K])
  Else Writeln('таких нет');
```

4.8. Поиск элементов массива, которые встречаются только один раз.

```
For I:=1 to KOL do
  begin
    NN:=0;
    For K:=1 to KOL do
      If Mas[I]=Mas[K] Then NN:=NN+1;
    If NN=1 Then Write(Mas[I]:3);
  end;
```

4.9. Поиск суммы элементов массива, кратных заданному числу

```
S:=0;  
WriteLn('Введите число ');  
ReadLn(K);  
For I:=1 To KOL Do  
    If Mas[I] mod K Then S:=S+Mas[I];
```

5. Записать в массив В номера ненулевых элементов массива MAS

```
J:=0;  
For I:=1 To KOL Do  
If Mas[I]<>0 Then Begin Inc(J);B[J]:=I End;
```

6. Копирование (перезапись одного массива в другой)

```
For I:=1 To KOL Do  
  B[I]:=Mas[I];  
  
если массивы идентичны, то B := Mas;
```

7. Численные оценки (сколько в массиве отрицательных элементов)

```
Kol_S:=0;  
For I:=1 To KOL Do  
  If A[I]<0 Then Kol_S:=Kol_S+1;
```

8. Перестановки элементов (переставить местами максимальный и минимальный элементы)

```
Max:=Mas[1];Min:=Mas[1];N_max:=1;N_min:=1;  
For I:=1 To KOL Do  
Begin  
  If Mas[I]>Max Then  
    Begin Max:=Mas[I];N_max:=I End;  
  If Mas[I]<Min Then  
    Begin Min:=Mas[I];N_min:=I End;  
End;  
Rab:=Mas[N_max];  
Mas[N_max]:=Mas[N_min];  
Mas[N_min]:=Rab;
```